
Zusammenfassung 

Eine mächtige, aber wenig bekannte Option in PROC GENMOD ist 
die OFFSET-Option. Sie wird in der SAS-Dokumentation für PROC 
GENMOD eingeführt, um in Log-Linearen-Modellen den Einfluss der 
Anzahl der Versuche ni auf die Anzahl der Erfolge yi zu berück-
sichtigen.  
 
Die OFFSET-Option eignet sich hervorragend, um bei Modellen, bei 
denen z.B. der Einfluss von Kovariablen durch externe Informationen 
vorgegeben ist, die freien Parameter in Abhängigkeit dieser Vorgaben 
zu optimieren. Aus diesem Ansatz leiten sich heuristische Methoden 
zur Modellauswahl und –Diagnose ab, auf die abschließend kurz ein-
gegangen wird. 
 

Keywords: Generalisiertes Lineares Modell, PROC GENMOD, Offset, a-
priori-Information, Parameter, Restriktionen. 

1 Einleitung 

Die Familie der Generalisierten Linearen Modelle (GLMs) wurde erstmalig 
von Nelder & Wedderburn [1972] zusammenhängend dargestellt. Wie der 
Name andeutet, stellen GLMs eine Erweiterung des klassischen linearen 
Modells dar. Viele bekannte Modelle, wie das Logit-, Probit- oder loglineare 
Modell, gehören zu der Familie der GLMs. Auf eine Einführung der theore-
tischen Grundlagen von GLMs wird an dieser Stelle verzichtet. Hier sei auf 
die einschlägige Literatur, wie z.B. die SAS-Dokumentation für PROC 
GENMOD verwiesen.  

Mit der PROC GENMOD bietet SAS ein mächtiges Tool nicht nur zur Mo-
dellierung einzelner GLMs, sondern vor allem zur Modellauswahl bzw. zum 
Modellvergleich. Eine i.d.R. wenig beachtete Modelloption ist die OFFSET-
Option. Mit Hilfe dieser Option können extern vorgegebene Effekte („a-pri-
ori-Informationen“) bei der Modellierung bzw. Schätzung der verbleibenden 
freien Parameter explizit berücksichtigt werden. 

2 Problemstellung 

In der praktischen Anwendung von GLMs -z.B. für die Tarifkalkulation in 
der Versicherungswirtschaft- müssen häufig externe Vorgaben für einzelne 
Merkmale bzw. Modellvariablen berücksichtigt werden. Ein typisches Bei-
spiel ist das Merkmal „Schadenfreiheitsrabatt“ in der Kraftfahrt-Haftpflicht-
Versicherung. Die Indexwerte („Modell-Parameter“) für dieses Merkmal 
sind in den Versicherungsbedingungen fest verankert und werden i.d.R. 
nicht mit jeder neuen Tarifgeneration angepasst.  
Diese vorgegebenen Modellparameter müssen natürlich bei der Ermittlung 
aller andern Parameter a priori berücksichtigt werden. Hierbei handelt es 
sich um eine Modellschätzung mit Restriktionen in den Parametern. Die 
prinzipielle Idee ist, a-priori-Informationen in geeigneter Weise vom Beo-
bachtungsvektor y der zu erklärenden Variable zu subtrahieren und die 
Modellschätzung für die verbleibenden Parameter mit dem modifizierten 
Beobachtungsvektor y* vorzunehmen. Auf diesem Ansatz basiert z.B. auch 
der verallgemeinerte F-Test zum Test auf eine Untermenge der Parameter. 

3 A-priori-Informationen im klassischen linearen 
Modell 

Einfach und anschaulich lässt sich der Ansatz zur Berücksichtigung von a-
priori-Informationen in der Modellschätzung am Beispiel des klassischen 
linearen Modells der Form 

 )....1(und),0(~mit' 2 TtNeey ttttt =+= σβx  (1)  

darstellen. Hierbei bezeichnen yt die Responsevariable bzw. xt den Vektor 
der erklärenden nicht stochastischen Variablen für die t-te der insgesamt T 
Beobachtungen und β  den zu schätzenden Parametervektor mit insgesamt P 

Elementen. Es wird angenommen, dass die Störterme et unabhängig normal-
verteilt sind. Das Modell wird in der Form  
 ]1[2]1[12211 undmit'' xNxMtttt ey βββxβx ++=  (2) 

aufgeteilt, so dass die a-priori-Informationen in dem Vektor 2β  zusammen-
gefasst werden. Alle verbleibenden und noch zu schätzenden Parameter sind 
in dem Vektor 1β  zusammengefasst. Nun werden die a-priori-Informationen 
auf beiden Seiten der Modellgleichung subtrahiert. 

 [ ]22
**

11
* 'mit' βxβx tttttt yyey −=+=  (3)  

Der Parametervektor 1β  des modifizierten Modells kann z.B. mit dem 
Kleinste-Quadrate-Schätzer  

 ( ) *
1

1
111 y'XX'Xˆ ΩΩ= −β  (4) 

geschätzt werden. In der Diagonalmatrix  Ω  sind die Beobachtungsgewichte 
zusammengefasst. Im klassischen Fall mit identischen Beobachtungsge-
wichten hat die Matrix  Ω  keinen Einfluss auf die Modellschätzung. 

 M
NMTNMT

M F
−−−−

− ~
)/('

/)''( **

ee
eeee

 (5) 

Auf der Logik aus (2) und (3) basiert auch der allg. F-Test (5) zum Test auf 
eine Untermenge der Parameter. Im Vektor 2β  sind Parameter, über die 
Hypothesen aufgestellt werden, zusammengefasst. 

4 A-priori-Informationen im generalisierten linearen 
Modell 

Die Verallgemeinerung dieses Ansatzes zur Berücksichtigung von a-priori-
Informationen in der Modellschätzung für GLMs ist etwas aufwendiger. Das 
GLM in der Form 

 ( ) ),()(mit'1
t

t
ttt V

n
yVaregy µφ

=+= − βx  (6) 

wobei g-1(..) die Linkfunktion und V(..) die Varianzfunktion des jeweiligen 
GLMs bezeichnen, wird wiederum so aufgeteilt, dass die a-priori-Informa-
tionen in dem Vektor 2β  zusammengefasst werden: 

Alle verbleibenden und noch zu schätzenden Parameter sind in dem Vektor 
1β  zusammengefasst. Hierbei bezeichnen nt die Beobachtungsgewichte und 

φ  den i.d.R. unbekannten Dispersionsparameter. Bei der Bildung des 
modifizierten Modells müssen die die a-priori-Informationen über die Link-
funktion auf das Skalenniveau von yt transformiert werden: 

Zu beachten ist, dass die Varianzfunktion sich nicht ändert. Sie bezieht sich 
weiterhin auf die untransformierte Responsevariable yt. Diese nur marginale 
Änderung in der Schätzung verhindert aber eine direkte Schätzung des modi-
fizierten Modells mit PROC GENMOD.  

5 Modellierung von GLMs mit PROC GENMOD 

Die Umsetzung in SAS wird an Hand des Datensatzes von Baxter, Coutts & 
Ross, (1980) demonstriert. Dieser Datensatz enthält 6.761 Schadenfälle mit 
Schadenzahlungen, die durch 3 Tarifmerkmale mit 5, 4 bzw. 3 Ausprägun-
gen in 60 Zellen zusammengefasst werden. Der Schadenbedarf „SB“, der 
sich aus dem Quotient der Schadenaufwendungen „c“ und der Jahreseinhei-
ten „je“ ergibt, wird durch drei Klassifikationsmerkmale „SFR“, „KW“ und 
„REGIO“ beschrieben. Ausgehend von dem Poisson-Modell mit logarithmi-
scher Linkfunktion wird in mehreren Schritten gezeigt, wie die Berücksich-
tigung von a-priori-Informationen in PROC GENMOD implementiert wer-
den kann. 
Bei der Modellierung mit PROC GENMOD (Beispiel 1) müssen neben der 
aus PROC GLM bekannten MODEL-Anweisung mit Angabe der Modellva-
riablen lediglich die Verteilung der Responsevariablen und die Linkfunktion 
(DIST- und LINK-Option in der MODEL-Anweisung) definiert werden. Mit 
der CLASS-Anweisung werden nominalskalierte Klassifikationsmerkmale 
definiert. Mit der WEIGHT- oder SCWGT-Anweisung wird die Gewich-
tungsvariable n berücksichtigt. Mit der DSCALE-Option wird der Disper-

sionsparameter φ , der im Poisson-Modell eigentlich per Definition eins ist, 
zusätzlich aus den Daten geschätzt. 

/*****************************************/ 
/**   Bsp. 1 Poisson-Modell “Einfach”   **/ 
/*****************************************/ 
proc genmod data=KSFE_daten; 
    class SFR KW REGIO ; 
    model SB =  SFR KW REGIO  
          / dist=poisson link=log dscale; 
    weight JE; 
run; 

 Beispiel 1: PROC GENMOD-Code 
 
In Tab. 1 ist der SAS-Output für Beispiel 1, das für alle folgenden Beispiele 
als Referenzmodell dient, ausschnittsweise widergegebenen,. In den fol-
gende Beispielen werden a-priori-Informationen, die i.d.R. den SFR-Para-
metern aus Beispiel 1 entsprechen, für dass Merkmal „SFR“ vorgegeben. 
Bei korrekter Spezifikation ist zu erwarten, dass die Parameterschätzer für 
die andern beiden Merkmale mit denen aus dem Referenzmodell aus Bei-
spiel 1 zusammenfallen.  
 

     Analysis Of Parameter Estimates 
 
  Parameter      DF  Estimate  Std. Error   
 
    Intercept    1    5.1714   0.0894  
    SFR     S1   1    0.2230   0.0564  
    SFR     S2   1    0.1319   0.0339  
    SFR     S3   1   -0.1404   0.0442 
    SFR     S4   1   -0.0168   0.0344 
    SFR     S5   0    0.0000   0.0000  
    KW      K1   1   -0.4025   0.0370 
    KW      K2   1   -0.2586   0.0385  
    KW      K3   0    0.0000   0.0000   
    REGIO   R1   1    0.6760   0.0834  
    REGIO   R2   1    0.5871   0.0837  
    REGIO   R3   1    0.3255   0.0949  
    REGIO   R4   0    0.0000   0.0000  
 
    Scale        0   18.8606   0.0000  

 Tabelle 1: Parameterschätzer für Beispiel 1 

6 Manueller Offset in PROC GENMOD 

In Beispiel 2 wird gezeigt, wie a-priori-Informationen –hier die Parameter 
für das Merkmal SFR aus Beispiel 1- manuell in PROC GENMOD berück-
sichtigt werden können. Analog der in (6) - (8) skizzierten Vorgehensweise 
müssen diese vorgegebenen Werte entsprechend transformiert von der zu 
erklärenden Variable „SB“ abgezogen werden. Neben dieser modifizierten 
erklärenden Variablen „SB_off“ wird auch die modifizierte Gewichtungsva-
riable „JE_off“ benötigt. Mit ihr wird sichergestellt, dass in PROC 
GENMOD die sich aus der unmodifizierten Variable „SB“ ergebene 
Varianzfunktion berücksichtigt wird. 
 
Beim Aufruf von PROC GEMOD sind entspechend die modifizierten Vari-
ablen „SB_off“ und „JE_off“ zu berücksichtigen und die Variable „SFR“ 
aus der MODEL-Anweisung zu streichen. 
 

/**************************************/ 
/**    Offset für das Merkmal SFR    **/ 
/**************************************/ 
data KSFE_neu; 
set  KSFE_daten; 
 if  SFR  =  'S1'  then  off=  0.2230; 
 if  SFR  =  'S2'  then  off=  0.1319; 
 if  SFR  =  'S3'  then  off= -0.1404; 
 if  SFR  =  'S4'  then  off= -0.0168; 
 if  SFR  =  'S5'  then  off=  0; 
 SB_off  =  sb / exp(off); 
 JE_off  =  je * exp(off); 
run; 
proc genmod data=KSFE_neu; 
    class SFR KW REGIO ; 
    model SB_off = KW REGIO   
          /dist=p link=log dscale; 
    weight JE_off; 
run; 
Beispiel 2: Manuelle Berücksichtigung von a-priori-Informatonen 

Wie erwartet entsprechen die in Tabelle 2 zusammengefassten Schätzer für 
die verbleibenden Merkmale „REGIO“ und „KW“ exakt denen aus dem 
Referenzmodell. Die Unterschiede in den Varianz-Schätzern sind aus-
schließlich auf den durch den Gewinn von 4 Freiheitsgraden durch die Ein-



 Statistik 

111 

sparung des Merkmals „SFR“ geänderten Schätzer für den Dispersions-
parameter φ  zurückzuführen. 

      Analysis Of Parameter Estimates 
 
Parameter  DF Estimate StdErr 
 
Intercept  1  5.17142 0.083993 
KW  K1 1 -0.40249 0.035301 
KW  K2 1 -0.25859 0.036867 
KW  K3 0  0 0 
REGIO  R1 1  0.675963 0.08012 
REGIO  R2 1  0.587116 0.08033 
REGIO  R3 1  0.325549 0.09113 
REGIO  R4 0  0 0 
 
Scale  0  18.12065  
Tabelle 2: Parameterschätzer für Beispiel 2  
 

Diese Vorgehensweise ist umständlich, da die Modell-Modifikationen ent-
sprechend der jeweils gewählten Link- und Varianzfunktion angepasst wer-
den müssen. Mit der OFFSET-Option in der MODEL-Anweisung von 
PROC GENMOD ist hingegen eine elegante und bequeme Vorgabe von a-
priori-Informationen möglich. 

7 Die OFFSET-Option in PROC GENMOD 

In der Dokumentation für PROG GENMOD wird die OFFSET-Option in 
einem Zusammenhang eingeführt, der nicht automatisch den Rückschluss 
auf unser Problem zulässt. Das entsprechende Code-Segment im linken 
Block von Beispiel 3 wurde lediglich auf unsere Nomenklatur angepasst. Er 
produziert identischen Output wie der rechte Code-Block, der unserem Bei-
spiel 1 entspricht.  

Proc genmod 
data=KSFE_daten; 
  Class SFR KW REGIO 
  Model c = SFR KW REGIO 
     / dist  = poisson 
       link  = log 
       offset = ln; 
run; 

Proc genmod 
data=KSFE_daten; 
  Class SFR KW REGIO 
  Model c = SFR KW REGIO 
     / dist  = poisson 
       link  = log; 
  Weight je; 
run; 

Beispiel 3: OFFSET-Option in der SAS-Dokumentation 
 
In dem Beispiel aus der SAS-Dokumentation wird nicht der Schadenbedarf 
SB, sondern die Summe der Schadenaufwendungen c modelliert, wobei die 
Anzahl der Jahreseinheiten je als Skalierungsgröße berücksichtigt werden 
muss. Das mit (6) bzw. Beispiel 1 vergleichbare Referenzmodell  

 ( )βx 'exp t
t

t
t je

c
SB =⎥

⎦

⎤
⎢
⎣

⎡
=  (9) 

wird über folgendem Zwischenschritt  

 ( )βx 't
t

t

je
c

nl =⎥
⎦

⎤
⎢
⎣

⎡
 (10) 

so umcodiert, dass die Gewichtungsvariable jet aus dem Nenner der Res-
ponse-Variable auf die Seite der erklärenden Variable verschoben wird: 

 [ ] ( ) )ln('ln ttt jec += βx  (11) 

Diese Darstellung (11) entspricht exakt unserem modifizierten Modell  
 2211 '')( βxβx tttg +=µ  (12) 

aus (7). Der zweite Term auf der Seite der erklärenden Variablen aus (11) 
wird mit der OFFSET-Option vorgegeben. Mit der OFFSET-Option können 
also a-priori-Informationen auf dem Skalenniveau des linearen Prädiktors 
berücksichtigt werden. 

8 Anwendungsbeispiele für die OFFSET-Option 

Die a-priori-Informationen aus Beispiel 2 können über die OFFSET-Option 
im MODEL-Befehl wesentlich eleganter berücksichtigt werden. Die Offset-

Informationen in der Variable „off“ werden nur mit der OFFSET-Option 
eingeführt. Der Output entspricht exakt dem aus Beispiel 2 (vgl. Tab. 2). 
 

/**************************************/ 
/**    Offset für das Merkmal SFR    **/ 
/**************************************/ 
 
data KSFE_neu; 
set  KSFE_daten; 
 if  SFR  =  'S1'  then  off=  0.2230; 
 if  SFR  =  'S2'  then  off=  0.1319; 
 if  SFR  =  'S3'  then  off= -0.1404; 
 if  SFR  =  'S4'  then  off= -0.0168; 
 if  SFR  =  'S5'  then  off=  0; 
run;  
 
proc genmod data=KSFE_neu; 
    class SFR KW REGIO ; 
    model SB = KW REGIO   
       /dist=p link=log dscale offset=off; 
    weight JE; 
run;  
 
Beispiel 4: Beispiel 2 mit OFFSET-Option  

 
Das Merkmal „SFR“ darf nicht mehr als erklärende Variable mitgeführt 
werden. Da die a-priori-Informationen auf dem Skalenniveau des linearen 
Prädiktors vorgegeben werden, zieht –entgegen Beispiel 2– eine Änderung 
der Varianz- und Linkfunktion keine weiteren Anpassung der orgaben nach 
sich.  
Alternativ können auch a-priori-Informationen nicht für alle, sondern nur für 
eine Ausprägung des Klassifikationsmerkmals „SFR“ vorgegeben werden. 
Zu berücksichtigen ist, dass für diese Ausprägung (hier „S1“) kein Parameter 
geschätzt werden darf. Deshalb wird die Ausprägung von „S1“ auf die durch 
die Cornerpoint-Parametrisierung gegebene Referenzkategorie „S5“ umco-
diert.  

/********************************/ 
/*      Offset für SFR S1       */ 
/********************************/ 
 data KSFE_daten_neu; 
 set  KSFE_daten; 
   if SFR = 'S1' then do  
     off = 0.123; 
     SFR = 'S5'; 
   end; 
 run;  
proc genmod data=KSFE_neu; 
    class SFR KW REGIO ; 
    model SB = SFR KW REGIO   
         /dist=p link=log dscale offset=off; 
     weight JE; 
run;  
Beispiel 5: OFFSET für nur eine Ausprägung eines Merkmals  

 
Die geschätzten Parameter aus dem Beispiel 5 sind in Tab. 3 zusammenge-
fasst. Wie erwartet wird für die Ausprägung „S1“ kein Parameter geschätzt. 
Da der Offset-Wert für „S1“ mit 0,123 nicht dem entsprechenden Schätzer 
aus dem Referenzmodell entspricht, verschieben sich die Parameter für die 
andern Merkmale leicht.  
 

    Analysis Of Parameter Estimates 
 
Parameter  DF Estimate StdErr 
 
Intercept  1  5.1831 0.0909 
SFR S2 1  0.1166 0.0334 
SFR S3 1 -0.1556 0.0442 
SFR S4 1 -0.0320 0.0339 
SFR S5 0  0  0 
KW K1 1 -0.4037 0.0377 
KW K2 1 -0.2606 0.0392 
KW K3 0  0  0 
REGIO R1 1  0.6793 0.0852 
REGIO R2 1   0.5935 0.0854 
REGIO R3 1  0.3288 0.0969 
REGIO R4 0  0  0 
 
Scale  0 19.2568 0 
Tabelle 3: Parameterschätzer für Beispiel 5 

Ohne die Umcodierung von „S1“ auf die Referenzkategorie „S5“ würde für 
den Parameter S1 ein Wert von 0,100 –also die Differenz zwischen Offset-
Wert 0,1230 und Schätzwert ohne Offset von 0,2230– geschätzt werden. 
Alle andern Parameter würden wiederum exakt mit denen aus dem Refe-
renzmodell zusammenfallen. In diesem Fall wäre nur in die Parametrisierung 
und nicht –wie eigentlich beabsichtigt- in die Modellstruktur eingegriffen 
worden. 
 
/*************************/ 
/*      Schritt 1        */ 
/* Modell mit KW u REGIO */ 
/*************************/ 
 
proc genmod data=KSFE_daten; 
   class sfr kw regio; 
   model sb = kw regio 
        / dist=p link=log 
          obstats; 
   make 'obstats' out=obs_1; 
   weight je; 
run; 
data KSFE_neu 
 (rename=(xbeta=off)); 
   merge KSFE_daten 
      obs_1(keep=xbeta); 
run; 

/***********************/ 
/*      Schritt 2      */ 
/* Nur SFR,  
   Model 1= offset     */ 
/***********************/ 
proc genmod data=KSFE_neu; 
   class sfr kw regio; 
   model sb = sfr 
        / dist=p link=log  
          obstats 
offset=off; 
   make 'obstats' 
out=obs_2; 
   weight je;  
run; 
 

Beispiel 6: Beliebiges OFFSET für y 
 
 
Die Offset-Werte müssen sich nicht immer auf einzelne Parameter 
(-gruppen) beziehen. Es können z.B. auch Offset-Werte direkt für die abhän-
gige Variable y vorgegeben werden. In Beispiel 6 wird in Schritt 1 ein Mo-
dell nur mit den beiden Klassifikationsmerkmalen „KW“ und „REGIO“ 
geschätzt. Über die ODS-Optionen werden die Schätzer „XBETA“ für den 
linearen Prädiktor jeder einzelnen Beobachtung wieder an den Datensatz 
„KSFE_daten“ angespielt.  
 
In Schritt 2 wird ein Modell nur mit dem dritten Klassifikationsmerkmal 
„SFR“ geschätzt, wobei der lineare Prädiktor aus Schritt 1 als Offset vorge-

geben wird. In Tab. 4 sind die Parameterschätzer für beide Modelle aus Bei-
spiel 6 angegeben. 
 
 
 
/***********************/ 
/*       Schritt 1     */ 
/***********************/ 
 
Parameter Estimates 
 
Parameter DF  Estimate 
 
Intercept 1  5.1867 
KW K1 1 -0.4053 
KW K2 1 -0.2619 
KW K3 0  0  
REGIO R1 1   0.6826 
 
REGIO R2 1   0.6035 
REGIO R3 1   0.3335 
REGIO R4 0   0 
 
Scale   0  26.51660 
 

/*********************/ 
/*     Schritt 2     */ 
/*********************/ 
 
Parameter Estimates 
 
Parameter DF Estimate 
 
Intercept 1 -0.0227 
SFR S1 1  0.2213 
SFR S2 1  0.1311 
SFR S3 1 -0.1409 
SFR S4 1 -0.0173 
SFR S5 0  0 
 
Scale   0 17.9770 
 

Tabelle 4: Parameterschätzer für Beispiel 6 
 
Der zweite Schritt aus Beispiel 6 kann als GLM für die Residuen aus Schritt 
1 interpretiert werden. Diese Art der Residuenanalyse eignet sich hervorra-
gend, um über Indikatorvariablen –wie im Beispiel 6 das Merkmal „SFR“- 
systematische und noch nicht im Modell berücksichtigte Einflüsse anschau-
lich aufzuzeigen. Durch die Fixierung der Parameterschätzer aus Schritt 1 
über die OFFSET-Option weisen die Parameter aus Schritt 2 nur den iso-
lierten zusätzlichen Einfluss aus.  
Im Gegensatz zu dem üblichen Ansatz, die Indikatorvariable ohne Offset 
zusätzlich in das Basismodell aufzunehmen, beeinträchtigen Korrelationen 
bzw. Wechselwirkungen zwischen dem Basismodell und der Indikatorvari-
able die Ergebnisse bzw. die Schätzer für die Indikatorvariable nicht. Wei-
terhin sind diese Ergebnisse kompakter -und auch anschaulicher zu interpre-
tieren- als z.B. Residuen-Plots. 

9 Schlussbemerkung  

Im Bereich der generalisierten linearen Modelle stellt PROC GENMOD ein 
sehr vielseitiges und leistungsfähiges Werkzeug dar. Verschiedene Facetten 
zur Berücksichtigung von a-priori-Informationen wurden theoretisch und in 
der Umsetzung mit Hilfe der OFFSET-Option in PROC GENMOD aufge-
zeigt.  
Über die Berücksichtigung von a-priori-Informationen kann auch der Ein-
fluss einzelner Variablen in einem Modell isoliert werden. Aus dieser Über-
legung leiten sich anschauliche heuristische Ansätze zur Modellauswahl und 
Residuenanalyse ab, auf die kurz eingegangen wurde. 

